Avtokorelacijska funkcija
(Avtor: Martin Horvat, november 2005)

■ Preambula

In[173]:=
```
Off[General::spell1];
```

■ Uvod

Kratek sestavek je posvecen prestavitvi izracunu avtokokorelacijske funkcije za vzorec velikosti N za nek signal $\{x_n: n=0,..,N-1\}$ in vplivu peridicnosti signala na njo. Za nek splosen signal tipicno privzamemo naslednjo obliko korelacijske funckije:

$$C(n) = \frac{1}{N}\sum_{k=0}^{N-1} x_k\, x_{k+n}, \qquad (1)$$

kjer ze moremo privzeti peridicnost signala $x_{n+N} = x_n$. Definicijo korelacije implementiramo s FFT kot :

In[174]:=
```
Clear[Corr];

Corr[seznam_] := InverseFourier[
    Abs[Fourier[seznam, FourierParameters → {1, -1}]]^2,
    FourierParameters → {1, -1}] / Length[seznam];
```

in v tem primeru imamo casovno zahtevnost izracuna O(N log(N)). V primeru, da je signal slabo koreliran, korelacija C(n) (1) relaksira h kvadratu povprecnega signala $< x >^2$, kjer je

$$< x > = \frac{1}{N}\sum_{k=0}^{N-1} x_k .$$

Tukaj je zanimov predvsem hitrost relakacije (upada) korelacije in zato uporabljamo reskalirano obliko avtokorelacijo:

$$C'(n) = \frac{C(n) - < x >^2}{C(0) - < x >^2}, \qquad (2)$$

V imenovalcu izraza (2) lahko prepoznamo standardno deviacijo signala:

$$\sigma^2 = C(0) - < x >^2 = \frac{1}{N}\sum_{k=0}^{N-1} (x_k - < x >)^2, \qquad (3)$$

In[176]:=
```
Clear[CorrDecaying]
CorrDecaying[seznam_] := Module[{m, A, mean, sigma2},
   m = Fourier[seznam, FourierParameters → {1, -1}];
   mean = m[[1]] / Length[seznam]; m[[1]] = 0;
   A = InverseFourier[Abs[m]^2, FourierParameters → {1, -1}] / Length[seznam];
   sigma2 = A[[1]];
   {mean, sigma2, A / sigma2}
  ];
```

Definicija (1) privzame periodicnost signala s periodo enako velikosti vzorca N. Ta se morda v primeru periodicnih signalov se neujema s periodo signala recimo m, torej m ne deli N. In torej pri periodicnosti ali pocasno padajoci (avto)korelaciji podatkov lahko v definiciji (1) lahko naredimo precejsnjo napako. Zato pa v tem primerih uporabimo definicijo korelacije, ki je "fizikalno" pravilna in nima stranskih efektov ob potencialni periodicnosti podatkov:

$$C(n) = \frac{1}{N-n} \sum_{k=0}^{N-n-1} x_k\, x_{k+n},$$

Za izracun C(n) za vrednosti indeksa n = 0,.., L−1 je casovna zahtevnost te metode O(L*N),

In[178]:=
```
Clear[CorrSlow];
CorrSlow[seznam_, len_] := Table[
   Sum[seznam[[n + k]] seznam[[n]], {n, 1, Length[seznam] - k}] / (Length[seznam] - k),
   {k, 0, len}];
```

vendar se da tudi pospesit s FFT,vendar z enim trikom.Uvedimo nov nabor−− vektor y velikosti 2N,ki je periodicno zakljucen:

$$y_n = x_n \text{ za } n = 0, .. N-1, \quad y_{n+N} = 0 \text{ za } n = 0, .. N-1 \text{ in } y_{n+2N} = y_n$$

in potem se lahko avto−korelacija napise kot:

$$C(n) = \frac{1}{N-n} \sum_{k=0}^{2N-1} y_k\, y_{k+n}, \qquad n = 0, .. N-1$$

kar se pa lahko seveda izracuna z FFT kot:

In[233]:=
```
Clear[CorrFast, seznam];
CorrFast[seznam_] := Module[{y} ,
   y = Take[InverseFourier[
      Abs[Fourier[Join[seznam, Table[0, {Length[seznam]}]],
         FourierParameters → {1, -1}]]^2, FourierParameters → {1, -1}
     ],
     Length[seznam]];
   Table[y[[i]] / (Length[seznam] - i + 1), {i, Length[seznam]}]
  ];
```

Casovna zahtevnost teh izracunov je O(2N Log(2N)), torej za neke tipicne vrednosti L je to boljsa moznost.

■ Rutine za pomoc pri prikazu podatkov

In[182]:=
```
<< Graphics`Graphics`
```

In[183]:=
```
Clear[MyListPlot, MyLogListPlot, MyListPlotWithDots, ShowDataGraphs1];

MyListPlot[l_List, xlabel_, ylabel_, title_, options__] := Module[{},

  ListPlot[l,
   PlotRange → All,
   DisplayFunction → Identity,
   AxesLabel → {xlabel, ylabel},
   PlotLabel → title,
   ImageSize → 400,
   TextStyle → {FontFamily → "Times", FontSize → 10},
   options]
 ]

MyLogListPlot[l_List, xlabel_, ylabel_, title_, options__] := Module[{},

  LogListPlot[l,
   PlotRange → All,
   DisplayFunction → Identity,
   AxesLabel → {xlabel, ylabel},
   PlotLabel → title,
   ImageSize → 400,
   TextStyle → {FontFamily → "Times", FontSize → 10},
   options]
 ]

MyListPlotWithDots[l_List, xlabel_, ylabel_, title_] := Module[{},

  Show[
   MyListPlot[l, xlabel, ylabel, title, PlotJoined → True],
   MyListPlot[l, xlabel, ylabel, title, PlotStyle → PointSize[0.02]]
  ]
 ]

ShowDataGraphs1[seznam_, corr_] := Show[
   GraphicsArray[
    {MyListPlot[seznam, "n", "xn", "Signal", PlotJoined → False],
     MyListPlotWithDots[corr, "n", "C(n)", "Avtokorelacija"]},
    ImageSize → 550
   ]
  ];
```

■ Enostaven periodicen signal

Imejmo enostaven periodicen signal velikosti M in generiran s preslikavo x_n = n mod 5 s periodo m = 5. Prvic izberemo velikost vzorca kot veckratnik periode M=5^k (a), drugic pa je M nekompatibilen z m (b).

In[188]:=
```
(* a *)
```

In[189]:=
```
m = 5;
M = 5^3;

seznam = Table[Mod[i, m], {i, M}];
corr = Corr[seznam];

ShowDataGraphs1[seznam, corr];
```





In[194]:=
```
(* b *)
```

In[195]:=
```
m = 5;
M = 128;

seznam = Table[Mod[i, m], {i, M}];
corr = Corr[seznam];

ShowDataGraphs1[seznam, corr];
```





■ Signal iz logisticne preslikave

Podatke signala iz logisticne preslikave $x_{n+1} = a\, x_n\,(1 - x_n)$ dobim s pomocjo algoritma:

In[200]:=
```
Clear[LogisticMap, a, x0, ntake, nskip];
LogisticMap[a_, x0_, nskip_, ntake_] := Module[{i, x = x0},
   Do [x = a * x (1 - x), {nskip}];
   Table[x = a * x (1 - x), {i, ntake}]
  ];
```

Poglejmo si signal (trajektorije) iz dolocenega okna, kjer obstajajo stabilne periodicne orbite v bifurkacijskem diagramu logisticne preslikave. Preizkusimo razlicne ideje glede izracuna avtokorelacijeske funckije za signal pri nekem parametru *a*.

In[202]:=
```
a = 3.9778;
Nskip = 10000;
Ntake = 6^3; (* lahko tudi vzamemo 6 x nekaj, recimo 180 *)

seznam = LogisticMap[a, 0.1, Nskip, Ntake];
corr = Corr[seznam];

ShowDataGraphs1[seznam, corr];
```





In[207]:=
```
a = 3.9778;
Nskip = 10000;
Ntake = 128;

seznam = LogisticMap[a, 0.1, Nskip, Ntake];
corr = Corr[seznam];

ShowDataGraphs1[seznam, corr];
```





In[253]:=
```
a = 3.9778;
Nskip = 10000;
Ntake = 6^3;
Nlen = 100;

seznam = LogisticMap[a, 0.1, Nskip, Ntake];

Timing[corr = CorrSlow[seznam, Nlen];]
Timing[corrX = CorrFast[seznam];]


ShowDataGraphs1[seznam, corr];
ShowDataGraphs1[seznam, Take[corrX, Nlen + 1]];
```

Out[258]=
```
{0.068004 Second, Null}
```

Out[259]=
```
{0.004 Second, Null}
```





Oglejmo si se signal (trajektorijo), ki pripada kaoticnemu rezimu logisticne preslikave:

In[223]:=

```
(* C *)
```

In[224]:=

```
a = 3.978;
Nskip = 10000;
Ntake = 2^20;

seznam = LogisticMap[a, 0.1, Nskip, Ntake];

corr = Corr[seznam];
corrdecay = CorrDecaying[seznam];

Show[
 GraphicsArray[
  {MyListPlot[Take[seznam, 1024], "n", "x_n", "Signal", PlotJoined → False],
   MyListPlot[Take[corr, 512], "n", "C(n)", "Ver. 1", PlotJoined → True],
   MyLogListPlot[Abs[Take[Last[corrdecay], 100]], "n", "|C'(n)|",
    "Ver. 2", PlotJoined → True]}, ImageSize → 550, GraphicsSpacing → 0.05
 ]
]
```



Out[230]=

- GraphicsArray -

Verjetno se lepo opazi plato korelacijske funkcije |C'(n)|, ki je sorazmeren z $N^{-1/2}$,

$$\lim_{n\to\infty} |C'(n)| \sim N^{-1/2},$$

in je artifakt povprecenja oz. koncne velikosti vzorca ($N < \infty$).